


# Advanced Circuit Analysis




with the HP-28S



## EduCALC

27953 CABOT RD.  
LAGUNA NIGUEL, CA 92677





**ADVANCED CIRCUIT ANALYSIS**

**With the HP-28S**

**by Robert R. Boyd**

**Copyright 1988**

**Robert R. Boyd**

**All rights reserved**

To my Wife LINDA and my Daughter JANA  
For still loving a husband/father  
who wasn't there.



## INTRODUCTION

The purpose of this work is to present a unique method of of circuit analysis developed by the author several years ago. However, without a computer the method was not of much practical value for circuits above 3rd order. With a calculator such as the HP-28S, there is a vehicle with which the method can be used on more complicated circuits.

Section I presents the method of setting up the circuit matrices by inspection. Merely by labeling the voltage nodes to be analyzed, and the branch impedances connected to the nodes, the node equations can be written using the principle of superposition. It is not necessary to identify loops, cut sets, trees, chords, links, etc., of the topology. One merely has to know how many impedances are connected the node in question, and which impedance is connected to the driving voltage. This information can be obtained just by inspection of the circuit.

With linear circuits that contain no dependent sources, the symmetric coefficient matrices using loop or node analysis are very easy to set up. This mnemonic method is presented in just about every undergraduate text on network analysis. However, when dependent sources are introduced, the symmetry disappears along with the mnemonic method. With the technique presented here, the symmetry of the coefficient matrix is of no concern.

Other HP calculators/computers can be used with the material in Section I. A good choice would be an HP-71B with the Math Pac. The HP-41CV or HP-41CX with the Advantage module will also work. Of course, any computer that has complex matrix and double precision capabilities can be used as well.

Section II is exclusively for those readers with HP-28S calculators and familiarity with the operation of the calculator is assumed. Most of the circuits given in section I are analyzed with complete descriptions of the main programs. No attempt has been made to minimize the program code. The interested reader will probably see a better way to do it.

A PC AT or XT with P-Spice has some advantages over writing out the node equations and setting up the circuit matrices, no matter how easy or systematic the information herein makes it. There are also analog design workstations appearing on the market that are absolutely fantastic in their capabilities. However, a PC with P-Spice will run about \$5,000 while the analog workstations are going for over \$50,000. The HP-28S sells for under \$200. Comparing capability per dollar, the method presented here using the HP-28S wins hands down.

## TABLE OF CONTENTS

TITLE	PAGE
Section I - Setting up The Circuit Matrices	1
Example 1 - Solve for $V_o$ by Superposition ( $N = 3$ )	1
Example 2 - Solve for $V_o$ by Superposition ( $N = 4$ )	3
Example 3 - Ladder Network	4
Example 4 - Lattice Network	6
Example 5 - Twin-T Network	8
Example 6 - Collector Feedback	9
Example 7 - CE Hybrid $\pi$ Transistor Model	11
Example 8 - Inverting Op-Amp	12
Example 9 - Adjustable Gain Differential Ampl	13
Example 10 - Non-Linear Circuits	15
Example 11 - Fifth Order Active Filter	16
Example 12 - Complementary Feedback Amplifier	18
Alternate Formulations	21


Section II - HP-28S Programs	22
Example 3 - Ladder Network (DC)	23
Example 3 - Ladder Network (AC)	26
Example 5 - Twin-T Network	27
Example 7 - CE Hybrid Pi Transistor Model	29
Example 8 - Inverting Op-Amp	31
Example 9 - Adjustable Gain Differential Ampl	32
Example 11 - Fifth Order Active Filter	33
Example 12 - Complementary Feedback Ampl	35
Example 10 - Non-linear Circuit	37
Appendix - I. Ladder Network Analysis	40
II. Building Branch Impedances with the HP-28S	43
III. Floating Voltage Sources	43
IV. Designing with K Factors	45
References	47

## SECTION I. Setting up The Circuit Matrices.

Notational convention:  $E \Rightarrow$  Independent voltage source  
 $V \Rightarrow$  Dependent voltage source or node voltage  
 $I \Rightarrow$  Dependent current source or node current  
 $Z \Rightarrow$  real or complex impedance

Two simple star networks will be analyzed to show a method of solving for the voltage at the center node that is easy to remember. This topology is chosen since any circuit can be formed by combining star networks with 2 or more branches. The method is then generalized to an N-branch star network.

Example 1 Solve for  $V_o$  by superposition:  
 $(N = 3)$



a. Set  $E_2 = 0$ ,  $V_o \rightarrow V_o'$ :

$$V_o' = \frac{E_1 Z_2 // Z_3}{Z_1 + Z_2 // Z_3} = \frac{E_1}{1 + \frac{Z_1}{Z_2 // Z_3}}$$

$$\text{Since } Z_2/Z_3 = \frac{1}{\frac{1}{Z_2} + \frac{1}{Z_3}}$$

$$V_0' = \frac{E_1}{1 + Z_1 \left( \frac{1}{Z_2} + \frac{1}{Z_3} \right)}$$

b. Set  $E_1 = 0$ ,  $V_0 \rightarrow V_0''$ :

$$V_0'' = \frac{E_2 Z_1/Z_3}{Z_2 + Z_1/Z_3} = \frac{E_2}{1 + \frac{Z_2}{Z_1/Z_3}}$$

$$V_0'' = \frac{E_2}{1 + Z_2 \left( \frac{1}{Z_1} + \frac{1}{Z_3} \right)}$$


Then by the superposition principle,

$$\begin{aligned} V_0 &= V_0' + V_0'' = \frac{E_1}{1 + Z_1 \left( \frac{1}{Z_2} + \frac{1}{Z_3} \right)} + \frac{E_2}{1 + Z_2 \left( \frac{1}{Z_1} + \frac{1}{Z_3} \right)} \\ &= E_1 K_1 + E_2 K_2, \end{aligned}$$

where  $K_1$  and  $K_2$  are dimensionless constants determined by  $Z_1$ ,  $Z_2$ , &  $Z_3$ , and are always  $< 1$ .

We try a second example to see if there is a consistent pattern occurring.

Example 2 Solve for  $V_o$  by superposition:  
( $N = 4$ )



$$V_o = \frac{E_1 Z_2 // Z_3 // Z_4}{Z_1 + Z_2 // Z_3 // Z_4} + \frac{E_2 Z_1 // Z_3 // Z_4}{Z_2 + Z_1 // Z_3 // Z_4}$$

$$+ \frac{E_3 Z_1 // Z_2 // Z_4}{Z_3 + Z_1 // Z_2 // Z_4} + \frac{E_4 Z_1 // Z_2 // Z_3}{Z_4 + Z_1 // Z_2 // Z_3}$$

$$V_o = \frac{E_1}{1 + \frac{Z_1}{Z_2 // Z_3 // Z_4}} + \frac{E_2}{1 + \frac{Z_2}{Z_1 // Z_3 // Z_4}}$$

$$+ \frac{E_3}{1 + \frac{Z_3}{Z_1 // Z_2 // Z_4}} + \frac{E_4}{1 + \frac{Z_4}{Z_1 // Z_2 // Z_3}}$$

$$V_o = \frac{E_1}{1 + Z_1 \left( \frac{1}{Z_2} + \frac{1}{Z_3} + \frac{1}{Z_4} \right)} + \frac{E_2}{1 + Z_2 \left( \frac{1}{Z_1} + \frac{1}{Z_3} + \frac{1}{Z_4} \right)}$$

$$+ \frac{E_3}{1 + Z_3 \left( \frac{1}{Z_1} + \frac{1}{Z_2} + \frac{1}{Z_4} \right)} + \frac{E_4}{1 + Z_4 \left( \frac{1}{Z_1} + \frac{1}{Z_2} + \frac{1}{Z_3} \right)}$$

$$V_o = E_1 K_1 + E_2 K_2 + E_3 K_3 + E_4 K_4.$$

Now the pattern is evident. In general:


$$V_o = \sum_{i=1}^N \frac{E_i}{1 + \text{impedance in series with } E_i \left( \begin{array}{l} \text{Sum of remaining reciprocal} \\ \text{impedances (admittances)} \\ \text{connected to node } V_o \end{array} \right)}$$

$$= \sum_{i=1}^N E_i K_i.$$

As shown in the next example, the  $E_i$ 's can be a mix of  $E_i$  and  $V_i$ .

Now we will use this method to write node equations by inspection.

Example 3 Ladder network.



Solve for dependent node voltages  $V_1$  and  $V_2$ .

We see that the equation for  $V_1$  will be the superposition sum of  $E_1$  and  $V_2$ :

$$(1) \quad V_1 = E_1 K_1 + V_2 K_2, \text{ where} \quad K_1 = \frac{1}{1 + Z_1 \left( \frac{1}{Z_2} + \frac{1}{Z_3} \right)},$$

$$K_2 = \frac{1}{1 + Z_3 \left( \frac{1}{Z_1} + \frac{1}{Z_2} \right)}$$

Since there is only one voltage driving node  $V_2$ , the equation for  $V_2$  will have only one term:



$$(2) \quad V_2 = V_1 K_2, \text{ where } K_2 = \frac{1}{1 + \frac{Z_2}{Z_1}}$$

Equations (1) and (2) can be solved for  $V_1$  and  $V_2$  by elimination or by matrix methods. By elimination:

$V_1 = E_1 K_1 + (V_1 K_2) K_2$ , substituting (2) into (1). Then

$$(3) \quad V_1 = \frac{E_1 K_1}{1 - K_2 K_2}$$

Substituting (3) into (2) gives  $V_2$ , or

$$V_2 = V_1 K_2 = \frac{E_1 K_1 K_2}{1 - K_2 K_2}$$

Using matrix methods:

Rearranging (1) and (2) so that the independent terms are on the LH side:

$$E_1 K_1 = V_1 - V_2 K_2$$


$$0 = V_2 - V_1 K_2$$

From this form it is easy to construct the coefficient matrix and independent column vector:

$$\begin{vmatrix} 1 & -K_2 \\ -K_2 & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \end{vmatrix} = \begin{vmatrix} E_1 K_1 \\ 0 \end{vmatrix}$$

From now on, matrices will be used exclusively. As the circuits get larger and more complicated, solving for the unknown node voltages using algebra and the elimination method becomes too lengthy and error prone.

Example 4 Lattice network.



Now we can be methodical and consistent.

Step 1. Write dependent node equations using superposition and assigning a unique K factor to each term.

$$V_1 = E_1 K_1 + E_2 K_2 + V_2 K_3$$

$$V_2 = E_1 K_4 + E_2 K_5 + V_1 K_6$$

(We don't care what the K's are until after the matrices are formed.)

Step 2. Put independent terms on LH side:

$$E_1 K_1 + E_2 K_2 = V_1 - V_2 K_3$$

$$E_1 K_4 + E_2 K_5 = V_2 - V_1 K_6$$

Step 3. Put in matrix form:

$$\begin{vmatrix} 1 & -K_3 \\ -K_6 & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \end{vmatrix} = \begin{vmatrix} E_1 K_1 + E_2 K_2 \\ E_1 K_4 + E_2 K_5 \end{vmatrix}$$

Step 4. From the circuit diagram and the equations of step 1, write out the K factors.

Before doing this, a functional notation for the K factors will be defined.

$$\text{Let } \frac{1}{1 + \frac{A}{B}} = F_2(A, B)$$

$$\frac{1}{1 + A \left( \frac{1}{B} + \frac{1}{C} \right)} = F3(A,B,C),$$

$$\frac{1}{1 + A \left( \frac{1}{B} + \frac{1}{C} + \frac{1}{D} \right)} = F4(A,B,C,D).$$

Note, for example, that  $F3(B,C,A) = \frac{1}{1 + B \left( \frac{1}{C} + \frac{1}{A} \right)}$

= F3(B,A,C), i.e., after the first variable, the order is not important since the reciprocals can be summed in any order.

(The F functions are not subscripted since this is the way they will appear in HP-28S programs.)

Getting back to Example 4, we can write out the K factors using this functional notation:

$$K1 = F3(Z1,Z2,Z4), \quad K2 = F3(Z4,Z1,Z2)$$


$$K3 = F3(Z2,Z1,Z4), \quad K4 = F3(Z5,Z2,Z3)$$

$$K5 = F3(Z3,Z2,Z5), \quad K6 = F3(Z2,Z3,Z5)$$

Remember that the first Z in F3 is in series with the E or V in question. For example, in  $V_i = E_i K_i + \dots$ , the first Z in  $K_i$  is between  $V_i$  and  $E_i$ , or  $Z_i$ . Be sure to account for all the remaining Z's connected to node  $V_i$ ; in this case  $Z_2$  and  $Z_4$ .

One last example before going on to transistor and op-amp circuits:

### Example 5 Twin-T Network



Step 1.

$$V_1 = E_1 K_1 + V_3 K_2$$

$$V_2 = E_1 K_3 + V_3 K_4$$

$$V_3 = V_1 K_5 + V_2 K_6$$

Step 2.

$$E_1 K_1 = V_1 - V_3 K_2$$

$$E_1 K_3 = V_2 - V_3 K_4$$

$$0 = V_3 - V_1 K_5 - V_2 K_6$$

Step 3.

$$\begin{vmatrix} 1 & 0 & -K_2 \\ 0 & 1 & -K_4 \\ -K_5 & -K_6 & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \\ V_3 \end{vmatrix} = \begin{vmatrix} E_1 K_1 \\ E_1 K_3 \\ 0 \end{vmatrix}$$

Step 4.


$$K_1 = F_3(Z_1, Z_2, Z_3), K_2 = F_3(Z_3, Z_1, Z_2),$$

$$K_3 = F_3(Z_4, Z_5, Z_6), K_4 = F_3(Z_6, Z_4, Z_5),$$

$$K_5 = F_3(Z_3, Z_6, Z_7), K_6 = F_3(Z_6, Z_3, Z_7).$$

The circuit is now ready for solution by the HP-28S.

At this point, we modify our basic star network by adding current sources:



Again, by superposition:


$$V_0 = E_1 K_1 + V_2 K_2 + V_3 K_3 + P_1 I_4 - P_1 I_5,$$

$$\text{where } K_1 = F_3(Z_1, Z_2, Z_3), K_2 = F_3(Z_2, Z_1, Z_3),$$

$$K_3 = F_3(Z_3, Z_1, Z_2), \text{ and } P_1 = Z_1 // Z_2 // Z_3.$$

Note direction of current flow and the sign attached;  
toward node => +, away from node => -.

Example 6 Collector Feedback.



Step 1.

$$V_1 = E_1 K_1 + V_2 K_2 - P_1 I_c$$

$$V_2 = E_2 K_2 + V_1 K_4 - P_2 I_b$$

$$V_3 = Z_3 I_e$$

$$V_2 - V_3 = V_{be} \quad (= 0.6V)$$

Step 2. (Express  $I_c$  and  $I_e$  in terms of  $I_b$ .)

$$E_1 K_1 = V_1 - V_2 K_2 + P_1 B I_b \quad (I_c = \text{Beta } I_b = B I_b)$$

$$E_2 K_2 = V_2 - V_1 K_4 + P_2 I_b$$

$$0 = V_2 - (1 + B) I_b Z_3 \quad [I_e = (1 + B) I_b]$$

$$V_{be} = V_2 - V_3$$

Step 3.

$$\begin{vmatrix} 1 & -K_2 & 0 & P_1 B \\ -K_4 & 1 & 0 & P_2 \\ 0 & 0 & 1 & -(1+B)Z_3 \\ 0 & 1 & -1 & 0 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \\ V_3 \\ I_b \end{vmatrix} = \begin{vmatrix} E_1 K_1 \\ E_2 K_2 \\ 0 \\ V_{be} \end{vmatrix}$$

Step 4.

$$K_1 = F_2(Z_1, Z_4), \quad K_2 = F_2(Z_4, Z_1),$$


$$K_3 = F_2(Z_2, Z_4), \quad K_4 = F_2(Z_4, Z_2)$$

$$P_1 = Z_1 // Z_4, \quad P_2 = Z_2 // Z_4$$

This circuit is not easily solved by conventional methods. Using the above matrices, the HP-28S will solve for all node voltages and the base current  $I_b$ . Collector and emitter currents are easily obtained from  $I_c = B I_b$ , and  $I_e = (1 + B) I_b$ .

Note that in forming the  $P$ 's associated with current sources, they are easily remembered as the parallel combination of all impedances connected to the node in question.

Example 7 Common Emitter Hybrid Pi Transistor Model



Note that  $(g_m V_1)$  is a voltage-controlled-current-source, or VCCS.

Step 1.

$$V_1 = E_1 K_1 + V_2 K_2$$

$$V_2 = V_1 K_3 - g_m V_1 P_1 = V_1 (K_3 - g_m P_1)$$

Step 2.

$$E_1 K_1 = V_1 - V_2 K_2$$

$$0 = V_2 - V_1 (K_3 - g_m P_1)$$

Step 3.


$$\begin{vmatrix} 1 & -K_2 \\ (g_m P_1 - K_3) & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \end{vmatrix} = \begin{vmatrix} E_1 K_1 \\ 0 \end{vmatrix}$$

Step 4.

$$K_1 = F_3(Z_1, Z_2, Z_3), \quad K_2 = F_3(Z_3, Z_1, Z_2)$$

$$K_3 = F_2(Z_3, Z_4), \quad P_1 = Z_3 / Z_4$$

### Example 8 Inverting op-amp



A simplified model of the op-amp is obtained by the dependent voltage source  $V_3 = -AV_1$ , where  $A$  is the open loop gain. ( $V_3$  is a voltage-controlled-voltage-source, or VCVS.)

The variable  $A$  can be complex to show the first order rolloff without adding additional reactive components:

$$A = \frac{A_{ol}}{1 + \frac{j\omega}{\omega_1}}, \text{ where } A_{ol} = \text{about } 10,000 \text{ v/v,}$$

and  $\omega_1 = 2\pi f_1$  is the frequency breakpoint, and  $10 < f_1 < 100$  in Hz for most op-amps.

Gain  $A$  can have zeros as well as poles:

$$\frac{A_{ol} \left( 1 + \frac{j\omega}{\omega_2} \right)}{\left( 1 + \frac{j\omega}{\omega_1} \right) \left( 1 + \frac{j\omega}{\omega_3} \right)}$$

For most op-amp circuits, the single pole rolloff will suffice.



Again, steps 1 thru 4 are no different:

Step 1.

$$V_1 = E_1 K_1 + V_2 K_2$$

$$\begin{aligned} V_2 &= V_1 K_3 + V_2 K_4 = V_1 K_3 - A V_1 K_4 \\ &= V_1 (K_3 - A K_4) \end{aligned}$$

Step 2.

$$E_1 K_1 = V_1 - V_2 K_2$$

$$0 = V_2 - V_1 (K_3 - A K_4)$$

Step 3.


$$\begin{vmatrix} 1 & -K_2 \\ (A K_4 - K_3) & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \end{vmatrix} = \begin{vmatrix} E_1 K_1 \\ 0 \end{vmatrix}$$

Step 4

$$K_1 = F_2(Z_1, Z_2), K_2 = F_2(Z_2, Z_1)$$

$$K_3 = F_3(Z_2, Z_3, Z_4), K_4 = F_3(Z_4, Z_2, Z_3)$$

Example 9 Adjustable gain differential amplifier.



Step 1.

$$V_1 = E_1 K_1 + V_2 K_2$$

$$V_2 = V_1 K_3 + V_3 K_4 + V_5 K_5$$

$$V_3 = A(V_4 - V_1)$$

$$V_4 = E_2 K_6 + V_5 K_7$$

$$V_5 = V_4 K_8 + V_2 K_9$$

Step 2.

$$E_1 K_1 = V_1 - V_2 K_2$$

$$0 = V_2 - V_1 K_3 - V_3 K_4 - V_5 K_5$$

$$0 = V_3 - A V_4 + A V_1$$

$$E_2 K_6 = V_4 - V_5 K_7$$

$$0 = V_5 - V_2 K_9 - V_4 K_8$$

Step 3.

$$\left| \begin{array}{ccccc|c|c|c} 1 & -K_2 & 0 & 0 & 0 & V_1 & & E_1 K_1 \\ -K_3 & 1 & -K_4 & 0 & -K_5 & V_2 & & 0 \\ A & 0 & 1 & -A & 0 & V_3 & = & 0 \\ 0 & 0 & 0 & 1 & -K_7 & V_4 & & E_2 K_6 \\ 0 & -K_9 & 0 & -K_8 & 1 & V_5 & & 0 \end{array} \right|$$

Step 4.

$$K_1 = F_2(R_1, R_2), K_2 = F_2(R_2, R_1)$$

$$K_3 = F_3(R_2, R_3, R_4), K_4 = F_3(R_3, R_2, R_4),$$


$$K_5 = F_3(R_4, R_2, R_3),$$

$$K_6 = F_2(R_5, R_6), K_7 = F_2(R_6, R_5),$$

$$K_8 = F_3(R_6, R_4, R_7), K_9 = F_3(R_4, R_6, R_7)$$

### Example 10 Non-linear Circuits.

Some non-linear diode circuits can be solved by converting the diodes to resistors (in series with a 0.6V source if need be). The method is to monitor the voltages across the resistor (diode) for polarity. If the "diode" becomes reverse biased, then change its value to 10 Megohms. If it becomes forward biased, change its value to, say, 10 ohms.



Step 1 (For the circuit on the right)

$$V_1 = E_1 K_1 + E_3 K_2 + V_3 K_3$$

$$V_2 = E_2 K_4 + E_3 K_5 + V_3 K_6$$

$$V_3 = V_1 K_7 + V_2 K_8$$

Step 2.

$$E_1 K_1 + E_2 K_2 = V_1 - V_2 K_3$$

$$E_2 K_4 + E_2 K_5 = V_2 - V_2 K_4$$


$$0 = V_2 - V_2 K_6 - V_1 K_7$$

Steps 3 and 4 are, as they say, "left as an exercise for the student".

During the analysis in section II, if  $(V_1 - V_2) < 0$ , set  $R_4 = 10$  ohms; if  $> 0$  set  $R_4 = 10$  Megohms. Similarly, if  $(V_2 - V_2) > 0$ , set  $R_7 = 10$  ohms; if  $< 0$  set  $R_7 = 10$  Megohms.

Examples 11 and 12 following illustrate the ease of writing node equations using the K method for relatively large and complicated circuits.

Example 11 Fifth Order Active Filter



Step 1.

$$V_1 = E_1 K_1 + V_2 K_2 + V_4 K_3$$

$$V_2 = V_1 K_4 + V_3 K_5$$

$$V_3 = V_2 K_6 + V_4 K_7$$

$$V_4 = V_1 K_8 + V_3 K_9 + V_4 K_{10}$$

$$V_5 = V_4 K_{11}$$

$$V_4 = A(V_2 - V_5)$$

Step 2.

$$E_1 K_1 = V_1 - V_2 K_2 - V_4 K_3$$

$$0 = V_2 - V_1 K_4 - V_3 K_5$$

$$0 = V_3 - V_2 K_6 - V_4 K_7$$

$$0 = V_4 - V_1 K_8 - V_3 K_9 - V_4 K_{10}$$

$$0 = V_5 - V_4 K_{11}$$

$$0 = V_4 - AV_2 + AV_5$$

Step 3.

$$\begin{array}{cccccc|c|c|c} 1 & -K_2 & 0 & -K_3 & 0 & 0 & V_1 & & E_1 K_1 \\ -K_4 & 1 & -K_5 & 0 & 0 & 0 & V_2 & & 0 \\ 0 & -K_6 & 1 & -K_7 & 0 & 0 & V_3 & & 0 \\ -K_8 & 0 & -K_9 & 1 & 0 & -K_{10} & V_4 & = & 0 \\ 0 & 0 & 0 & 0 & 1 & -K_{11} & V_5 & & 0 \\ 0 & 0 & -A & 0 & A & 1 & V_4 & & 0 \end{array}$$

Step 4.

$$K1 = F4(Z1, Z2, Z3, Z4), K2 = F4(Z3, Z1, Z2, Z4)$$

$$K3 = F4(Z4, Z1, Z2, Z3)$$


$$K4 = F3(Z3, Z5, Z6), K5 = F3(Z6, Z3, Z5)$$

$$K6 = F3(Z6, Z7, Z8), K7 = F3(Z7, Z6, Z8)$$

$$K8 = F3(Z4, Z7, Z9), K9 = F3(Z7, Z4, Z9)$$

$$K10 = F3(Z9, Z4, Z7), K11 = F2(Z10, Z11)$$

Example 12 Complementary Feedback Amplifier.



Step 1.

$$V_1 = E_1 - I_{e2}Z_1$$

$$V_2 = E_1K_1 + V_3K_2 + I_{b2}P_1$$

$$V_3 = V_2 - I_{c1}Z_3$$

$$V_4 = E_3 - I_{b1}Z_4$$

$$V_5 = E_2K_3 + V_4K_4 + I_{e1}P_2$$

$$V_4 = E_2K_5 + V_5K_6 + I_{c2}P_3$$

$$V_{be1} = V_4 - V_5$$

$$V_{be2} = V_1 - V_2$$

Step 2.

$$E_1 = V_1 + (1 + B_2)I_{b2}Z_1$$

$$E_1K_1 = V_2 - V_3K_2 - I_{b2}P_1$$

$$0 = V_3 - V_2 + B_1I_{b1}Z_3$$

$$E_3 = V_4 + I_{b1}Z_4$$

$$E_2K_3 = V_5 - V_4K_4 - (1 + B_1)I_{b1}P_2$$

$$E_2K_5 = V_4 - V_5K_6 - B_2I_{b2}P_3$$

$$V_{be1} = V_4 - V_5$$

$$V_{be2} = V_1 - V_2$$

Step 3.

$$\begin{array}{cccccccc|c}
 1 & 0 & 0 & 0 & 0 & 0 & 0 & (1+B_2)Z_1 & V_1 \\
 0 & 1 & -K_2 & 0 & 0 & 0 & 0 & -P_1 & V_2 \\
 0 & -1 & 1 & 0 & 0 & 0 & B_1Z_2 & 0 & V_3 \\
 0 & 0 & 0 & 1 & 0 & 0 & Z_4 & 0 & V_4 \\
 0 & 0 & 0 & 0 & 1 & -K_4 & -(1+B_1)P_2 & 0 & V_5 \\
 0 & 0 & 0 & 0 & -K_4 & 1 & 0 & -B_2P_3 & V_6 \\
 0 & 0 & 0 & 1 & -1 & 0 & 0 & 0 & Ib_1 \\
 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & Ib_2
 \end{array}$$

$$= \begin{bmatrix} E_1 & E_1K_1 & 0 & E_3 & E_2K_3 & E_2K_5 & Vbe_1 & Vbe_2 \end{bmatrix}^T$$

(The  $^T$  transposes the row vector into a column vector.)

Step 4.

$$K1 = F2(Z2,Z3), K2 = F2(Z3,Z2)$$

$$K3 = F2(Z5,Z7), K4 = F2(Z7,Z5)$$

$$K5 = F2(Z6,Z7), K6 = F2(Z7,Z6)$$

$$P1 = Z2//Z3, P2 = Z5//Z7, P3 = Z6//Z7$$

In setting up the coefficient matrix, advantage should be taken of all the zeros in the matrix (a so-called sparse matrix) and that the main diagonal is nearly all 1's. That is, one should form an identity matrix first, and then store only the non-zero elements. For example, the HP-28S sequence

```
... 8 IDN 'A' STO B2 1 + Z1 * 'A(1,8)' STO ...
```

will store the first nonzero element.



By now the astute reader has probably seen the similarity of the F3 and F4 K factor functions with the parallel impedance function. There are two alternate formulations of these functions that may result in shorter HP-28S programs.

For example, if

$$K1 = F4(Z1, Z2, Z3, Z4), \quad K2 = F4(Z2, Z1, Z3, Z4),$$

$$K3 = F4(Z3, Z1, Z2, Z4), \quad K4 = F4(Z4, Z1, Z2, Z3),$$

a shorter way of computing K1 through K4 is

$$K1 = F4(Z1, Z2, Z3, Z4), \quad K2 = (K1 * Z1) / Z2,$$

$$K3 = (K1 * Z1) / Z3, \quad K4 = (K1 * Z1) / Z4.$$

Note that the denominator of  $K_i$  is  $Z_i$ ,  $i = 2, 3, 4$ .

Still another way of calculating K1 through K4 is

$$P1 = Z1 // Z2 // Z3 // Z4, \text{ then}$$

$$K1 = P1 / Z1, \quad K2 = P1 / Z2, \quad K3 = P1 / Z3, \text{ and } K4 = P1 / Z4.$$

The form here is  $K_i = P1 / Z_i$

Also note that  $K1 + K2 + K3 + K4 = 1$ .

These methods should be used after becoming familiar with the HP-28S programming structures in section II. For the sake of consistency, the programs in section II will be given using the original method.

## SECTION II. HP-28S Programs.

We will begin from the bottom and work up. That is, the following subprograms will be used with a main program. They should be keyed-in now. The name is given in single quotes. The author uses a parent directory of 'CKTAS' for circuit analysis programs, and these programs should be in a directory by that name to follow the solutions given here.

-----  
'HZ' (Hz to radians to reactance)

<< F ALOG 2 pi \* \* \* >>  
-----

'C' (Capacitive reactance.  $0 - j/wC$ )

<< HZ INV NEG 0 SWAP R->C >>  
-----

'L' (Inductive reactance.  $0 + jwL$ )

<< HZ 0 SWAP R->C >>  
-----

'SRC' (Series R C branch impedance.  $R - j/wC$ )

<< HZ INV NEG R->C >>  
-----

'PRC' (Parallel R C branch impedance.  $\frac{1}{1/R + jwC}$ )

<< HZ SWAP INV SWAP R->C INV >>  
-----

'SRL' (Series R L branch impedance.  $R + jwL$ )

<< HZ R->C >>  
-----

'PRL' (Parallel R L branch impedance.  $\frac{1}{1/R - jwL}$ )

<< HZ NEG INV SWAP INV SWAP R->C INV >>  
-----

'F2' (K factor F2 function.

<< -> a b '1/(1+a/b)' >>  
-----

'F3'

<< -> a b c '1/(1+a\*(1/b+1/c))' >>  
-----

-----  
'F4'

<< -> a b c d '1/(1+a\*(1/b+1/c+1/d))' >>  
-----

'P22' (Two parallel impedances, e.g.,  $P1 = Z_1 // Z_2$ )

<< INV SWAP INV + INV >>  
-----

'P23' (Three parallel impedances)

<< P22 P22 >>  
-----

'P24' (Four parallel impedances)

<< P23 P22 >>  
-----

We will first solve example 3 (ladder network) using the DC analysis program DCAP given below. For this analysis  $Z_1$  thru  $Z_4$  will become  $R_1$  thru  $R_4$ .

Following this example, the AC analysis program ACAP will be demonstrated using capacitors for two of the branches.

The following short program DCAP should now be written in the CKTAS directory:

'DCAP'

<< KRV SKF MAT CKTAS >>

The four commands are explained as follows:

KRV (K and R Variables) is a subdirectory under the CKTAS parent directory. This subdirectory should be created before running 'DCAP'. Using a subdirectory to store the K factors and component values keeps alot of clutter out of the parent directory CKTAS.

SKF (Store K Factors) is a subprogram that is unique to each circuit. It is used to obtain the component values in the KRV subdirectory and create the K factors. The K factors are then stored under their names, i.e,  $K1 \rightarrow 'K1'$ , etc.

The values of  $R1 = 1K$ ,  $R2 = 2K$ ,  $R3 = 3K$ , and  $R4 = 4K$ , should now be stored under their names in the KRV subdirectory. Also store 10 in the variable 'E1'.

When in the KRV subdirectory, store the following:

'UP'

<< CKTAS >>

This is used to easily get back up to the CKTAS directory.

For example 2, SKF (in the CKTAS directory) will be:

'SKF'

<< 'F3(R1,R2,R3)' 'K1' STO [K1 created.]

'F3(R3,R1,R2)' 'K2' STO [K2 created]

'F2(R3,R4)' 'K3' STO >> [K3 created]

The subprogram MAT (matrix), also in the CKTAS directory, creates the coefficient matrix and the column vector and solves for the unknowns, in this case, V1 and V2. (See page 4.)

'MAT'

<< 1 K2 NEG [1st row of matrix]

K3 NEG 1 [2nd row]

2 DUP 2 ->LIST ->ARRY [Create the 2 x 2 matrix]

E1 K1 \* 0

2 1 ->LIST ->ARRY [Column vector]

SWAP / ARRY-> DROP [Solve for V1 & V2]

SWAP DROP >> [Drop V1]

Now run the program DCAP and you should see (V2 =) 3.478  
In level 1. As a check, K1 = 0.545, K2 = 0.182, and  
K3 = .571 in the KRV subdirectory.

The AC analysis program ACAP is a little more involved. However it will not change from circuit to circuit.

'ACAP'

<< 'BF' STO	[BF=beginning frequency]
INV 'PD' STO	[PD=points/decade]
BF + 'ND' STO	[ND=total no. of decades]
ND BF - PD / 1 + 1 ->LIST	
0 0 R->C CON 'VOUT' STO	[VOUT=storage for output]
'VOUT' 1 1 ->LIST	[Put indexed 'VOUT' on stack]
BF ND FOR f f 'F' STO	[Store log frequency]
DCAP	[See DCAP program]
R->P C->R	[Put magnitude & phase angle on stack]
IF 2 FS? THEN SWAP DROP	[Get phase angle if flag 2 set]
ELSE DROP LOG 20 * END	[Else get dB magnitude]
F SWAP R->C PUTI	[Put (F,dB) or (F,angle) in VOUT storage]
PD STEP	[Next frequency]
CLEAR VOUT ARRY-> DROP >>	[Clear and put VOUT on stack]

Note: The CKTAS directory should contain the following objects, not necessarily in the order given: DCAP KRV SKF MAT ACAP PZ4 PZ3 PZ2 F4 F3 F2 SRL PRL HZ PRC SRC C L. The KRV subdirectory should have only components and K factor values plus the directory change program 'UP'.

The listing for the Bode plot (BPLT) is as follows:  
(Stored in CKTAS directory.)

```
'BPLT'  
  
<< 'Y1' STO 'Y2' STO           [Store plot limits]  
BF Y1 R->C PMIN                 [Set plot parameters]  
ND Y2 R->C PMAX BF 0 R->C AXES  
CLLCD DRAX 'VOUT' 1 1 ->LIST    [Put 'VOUT' on stack]  
BF ND START GETI PIXEL          [Plot]  
PD STEP DROP >>                [Drop 'VOUT' index]
```

Now for the ACAP run, store the following values in the  
KRV subdirectory:

R1 = 10K, R3 = 10K, C2 = 0.01uF, C4 = 0.01uF

E1 = 1.

The subprogram MAT does not have to be changed.

Modify the 'SKF' subprogram as follows:

```
'SKF'  
  
<< C2 C 'Z2' STO C4 C 'Z4' STO 'F3(R1,Z2,R3)'  
'K1' STO 'F3(R3,Z2,R1)' 'K2' STO 'F2(R3,Z4)' 'K3' STO >>
```

To run ACAP, key in the number of decades ND, space,  
points per decade PD, space, and the log start frequency BF.  
Press ACAP and the program will start.

For example, examine 3 decades at 10 points per decade  
starting with 100 Hz. (Log 100 = 2) Key in:

3, space, 10, space, 2, then press ACAP

When finished, the top number on the stack (level 31) should be (2.000,-0.118) in FIX 3 format. The level 1 number should be (5.000,-71.935). To plot this run, clear the stack and enter 0 space -75 BPLT. The corner frequency breakpoint should be very evident at about  $\log F = 2.8$  or  $F = 631$  Hz. The second breakpoint will occur at  $F = 13090$  Hz or  $\log F = 4.117$ .

The slope at  $\log F = 5$  is the dB value at  $\log F = 4.9$  minus the dB value at  $\log F = 5$  divided by the frequency increment of  $1/ND$  or

$$\text{slope} = (-71.935 - (-67.939))10 = -39.96 \text{ dB/decade}$$

or approximately -40 dB/decade, which we would expect for a two pole low pass filter.

Pressing ATTN will clear the plot and you should see the name 'VOUT' at level 1. Pressing EVAL, ARRY->, DROP., will return the data points to the stack.

For new component values, neither SKF nor MAT has to be changed. For new component types such as Z1 ---> C1, Z2 ---> R2, Z3 ---> C3, and Z4 ---> R4, (a high pass filter), only SKF has to be changed. Of course, for a new circuit topology, e.g. twin-T, both SKF and MAT will have to be changed.

Example 4, the lattice network is omitted.

#### Example 5. Twin-T Network

Step 1. Before entering values for the twin-T network, the ladder network values must be purged from the KRV subdirectory: KRV, MEMORY, NEXT, CLUSR, ENTER, USER. You should see an empty (KRV) directory.

Step 2. Key in and store the following values:

C1 = 0.01uF  
 R2 = 133K  
 C3 = 0.01uF  
 R4 = 267K  
 C5 = 0.02uF  
 R6 = 267K  
 R7 = 10Meg  
 E1 = 1

Also key in << CKTAS >>, 'UP', STO, for the directory change. Press 'UP' when done to change directories.

Step 3. Now key in the following SKF and MAT subprograms for the twin-T network in the CKTAS directory: (See page 8)

'SKF'

```
<< C1 C 'Z1' STO C3 C 'Z3' STO C5 C 'Z5' STO
'F3(Z1,R2,Z3)' 'K1' STO 'F3(Z3,Z1,R2)' 'K2' STO
'F3(R4,Z5,R6)' 'K3' STO 'F3(R6,R4,Z5)' 'K4' STO
'F3(Z3,R6,R7)' 'K5' STO 'F3(R6,Z3,R7)' 'K6' STO >>
```

'MAT'

```
<< 1 0 K2 NEG 0 1 K4 NEG K5 NEG K6 NEG 1
3 3 2 ->LIST ->ARRY [Coeff. array]
K1 E1 * K3 E1 * 0 3 1 ->LIST ->ARRY [Indep vector]
SWAP / ARRY-> DROP 3 ROLLD [Get V3]
DROP2 >> [Drop V1 & V2]
```

Step 4.

The component values given are for a 60 Hz notch filter. Hence we want to look at one decade between 10 and 100 Hz. Twenty points should be enough, therefore key in:

1, space, 20, space, 1, press ACAP

When finished, level 1 should show (2.000,-11.777); level 5 should show the notch: (1.800,-31.184). Finally the top level of the stack (level 21) should show: (1.000,-2.046).

To get more resolution, we can do 40 points starting from 30 Hz. To do this key in:


1, space, 40, space, 30, (LOGS menu), LOG, USER, ACAP

The notch is very prominent at level 29: (1.777,-55.595) To plot this key in: 0, space, -60, press BPLT.



### Example 7 Common Emitter Hybrid Pi Transistor Model

The circuit used for the analysis is shown below:



Note that  $Z_1 \rightarrow R_1$ ;  $Z_2 \rightarrow R_2 // C_1$ ;  $Z_3 \rightarrow R_3 // C_2$ ;  $Z_4 \rightarrow R_4 // R_5$ . (See page 11.)

Step 1. Purge the contents of the KRV directory except 'UP'.

Step 2. Store the following typical component values in the KRV subdirectory: (See reference 1.)

$R_1 = 100$	$R_3 = 4 \text{ Meg}$
$R_2 = 1 \text{ K}$	$R_4 = 80 \text{ K}$
$C_1 = 100 \text{ pF}$	$R_5 = 10 \text{ K (Collector load)}$
$C_2 = 3 \text{ pF}$	$GM = 0.025 \text{ (25 mA/V)}$

Step 2. Modify the SKF and MAT subprograms as follows:

'SKF'

<< R2 C1 PRC 'Z2' STO R3 C2 PRC 'Z3' STO

R4 R5 PZ2 'Z4' STO R1 Z2 Z3 F3 'K1' STO

Z3 R1 Z2 F3 'K2' STO Z3 Z4 F2 'K3' STO Z3 Z4 PZ2

'P1' STO >>

(Note faster method of calling F2 & F3 functions.)

'MAT'

<< 1 K2 NEG GM P1 \* K3 - 1 2 2 2 ->LIST ->ARRY

K1 0 2 1 ->LIST ->ARRY SWAP / ARRY-> [E1 = 1]

DROP SWAP DROP >> [Drop V1]

Step 4. Analyze the circuit.

Examine the collector output for 5 decades, 5 points per decade, beginning with log F = 3 (1 KHz):

5, space, 5, space, 3, press ACAP

At the end of the run see:

Level 26: (3.000,46.045) Start

Level 10: (6.200,43.336) 3 dB rolloff point

Level 1: (8.000,5.109) Finish

Plot from +50 to 0 dB:

CLEAR, 50, space, 0, BPLT

The loss of gain at high frequency is very apparent.

The input impedance can be obtained from: (E1 = 1)

$$Z_{in} = \frac{E_1}{I_1} = \frac{1}{\frac{1 - V_1}{R_1}} = \frac{R_1}{1 - V_1}$$

The output impedance can be obtained from the open circuit output voltage divided by the short circuit current, or

$$Z_{out} = V_{oc}/I_{sc}.$$

This can be accomplished by setting R5 to some very small value, say, 0.01 ohms, dividing the new output voltage by this value to obtain I<sub>sc</sub>, and then dividing into V<sub>oc</sub> to obtain Z<sub>out</sub>. With R5 = 10 Meg in the analysis, the output obtained with this value represents V<sub>oc</sub>.

### Example 8 Inverting op-amp

Step 1. Purge KRV except for UP.

Step 2. Store the following component values.

```
R1 = 10 K           (Z1)
R2 = 15 K           (Z2)
R3 = 1 K; C1 = 0.015 uF (Z3)
R4 = 15 K           (Z4)
A0 = 10E3           (Aol)
W1 = 2 * pi * 10E6 = 62.832E6 (Op-amp rolloff)
E1 = 1
```

Step 3. Key in SKF and MAT:

'SKF'

<< R3 C1 SRC 'Z3' STO R1 R2 F2 'K1' STO

R2 R1 F2 'K2' STO R2 Z3 R4 F3 'K3' STO

R4 Z3 R2 F3 'K4' STO

1 1 HZ [ "1 HZ" gives w]

W1 / R->C A0 SWAP / 'A' STO >> [A = Aol/(1 + Jw/w1)]

'MAT'

<< 1 K2 NEG A K4 \* K3 - 1 2 2 2 ->LIST

->ARRY E1 K1 \* 0 2 1 ->LIST ->ARRY

SWAP / ARRY-> DROP2 A \* NEG >> [V3 = -A V1]

Step 4. Analyze the circuit.

Do five decades at 5 points per decade starting at 10 Hz:

5, space, 5, space, 1, ACAP

```
Level 26: (1.000,9.539)    [DC gain of 30K/10K in dBV]
Level 1:  (6.000,28.094)    [Gain increase due to feedback
                             T network.]
```

Plot from 5 to 30 dB:

CLEAR, 30, space, 5, BPLT

The plot confirms that there is a zero at about  $\log F = 3.2$  and a pole at about  $\log F = 4$ .

Now decrease W1 to a more realistic 62.832 (10 Hz) to see the affects of op-amp rolloff.

The plot limits of Y2 = 25 and Y1 = -25 should show the change.

Example 9 Adjustable Gain Differential Amplifier.  
(Reference 2.)

Step 1. Purge KRV except for UP.

Step 2. Store the following component values in KRV subdirectory:

R1 = 20 K  
R2 = 2 K  
R3 = 2 K  
R4 = 1 K  
R5 = 20 K  
R6 = 2 K  
R7 = 2 K  
A0 = 10000  
E1 = 10  
E2 = -10

(Note: Since this circuit is intended for DC applications, rolloff affects due to W1 will not be included.)

Step 3. Key in SKF & MAT as follows:

'SKF'

<< R1 R2 F2 'K1' STO R2 R1 F2 'K2' STO

R2 R3 R4 F3 'K3' STO R3 R4 R2 F3 'K4' STO

R4 R2 R3 F3 'K5' STO R5 R6 F2 'K6' STO

R6 R5 F2 'K7' STO R6 R4 R7 F3 'K8' STO

R4 R6 R7 F3 'K9' STO

```

A0 'A' STO >>
'MAT'
<< 5 IDN 'A1' STO K2 NEG 'A1<1,2>' STO
K3 NEG 'A1<2,1>' STO K4 NEG 'A1<2,3>' STO
K5 NEG 'A1<2,5>' STO A 'A1<3,1>' STO A NEG 'A1<3,4>' STO
K7 NEG 'A1<4,5>' STO K9 NEG 'A1<5,2>' STO
K8 NEG 'A1<5,4>' STO
E1 K1 * 0 0 E2 K6 * 0 5 1 ->LIST ->ARRY
A1 / ARRY-> 3 DROPN
3 ROLL D DROP2 >> [Get V3]

```

Step 4. Analyze the circuit.

Press DCAP to see: -11.993 V

Change R4 to 3 K; press DCAP to see -6.665.

V3 = -10.002 V when R4 = 1.332 K.

(Example 10 will be covered after example 12.)

#### Example 11 Fifth Order Active Filter

Step 1. Purge KRV except UP.

Step 2. Store the following values in the KRV subdirectory:

C1 = 0.03 uF	(21)
R2 = 2 K	(22)
R3 = 70 K	(23)
C4 = 0.02 uF	(24)
C5 = 1.895 nF	(25)
R6 = 140 K	(26)
C7 = 0.01 uF	(27)
R8 = 12 K	
C8 = 0.4 nF	(28 = R8//C8
R9 = 2.7 K	(29)
R10 = 3.2 K	(210)

R11 = 10 K	(Z11)
A0 = 10000	
W1 = 62.283	(10 Hz op-amp rolloff)

Step 3.

'SKF'

<< C1 C 'Z1' STO C4 C 'Z4' STO C5 C 'Z5' STO C7 C 'Z7'

STO R8 C8 PRC 'Z8' STO Z1 R2 R3 Z4 F4 'K1' STO

R3 Z1 R2 Z4 F4 'K2' STO Z4 Z1 R2 R3 F4 'K3' STO

R3 R6 Z5 F3 'K4' STO R6 R3 Z5 F3 'K5' STO

R6 Z7 Z8 F3 'K6' STO Z7 R6 Z8 F3 'K7' STO

Z4 Z7 R9 F3 'K8' STO Z7 Z4 R9 F3 'K9' STO

R9 Z4 Z7 F3 'K10' STO R10 R11 F2 'K11' STO

A0 1 1 HZ W1 / R->C / 'A' STO >>

'MAT'

<< K1 0 0 0 0 0

[Input vector

E1 = 1]

6 1 ->LIST ->ARRY

[Done 1st to save  
swap time]

1 K2 NEG 0 K3 NEG 0 0

[1st row]

K4 NEG 1 K5 NEG 0 0 0

[2nd row]

0 K6 NEG 1 K7 NEG 0 0

[3rd row]

K8 NEG 0 K9 NEG 1 0 K10 NEG

[4th row]

0 0 0 0 1 K11 NEG

[5th row]

0 0 A NEG 0 A 1

[6th row]

6 6 2 ->LIST ->ARRY

/ ARRY-> DROP 6 ROLLD

5 DROPN >>

[Get V6]

Step 4.

Due to its size, the ACAP program will take some time.  
(About 12 seconds for each frequency point.)

Examine 5 decades, 5 points per decade, starting at  
100 Hz:

5, space, 5, space, 2, ACAP

The stack should show:

Level 26: (2.000,-51.889)  
Level 18: (3.600,1.965) [Peak value]  
Level 1: (7.000,-40.515)

To plot: CLEAR, 5, space, -55, BPLT

The area between  $\log F = 2.0$  and  $\log F = 3.4$  shows an  
elliptical response with a very steep 5th order climb  
to the peak value, and a 1st order rolloff.

#### Example 12 Complementary Feedback Amplifier

Step 1. Purge KRV except for UP.

Step 2. Store the following component values in the KRV  
subdirectory.

R1 = 200	E1 = +15 V
R2 = 1K	E2 = -15 V
R3 = 0.1	E3 = +5 V
R4 = 10K	B1 = 100
R5 = 1.5K	B2 = 100
R6 = 300	
R7 = 5.1K	

Step 3.

'SKF'

<< R2 R3 F2 'K1' STO R3 R2 F2 'K2' STO

R5 R7 F2 'K3' STO R7 R5 F2 'K4' STO

R6 R7 F2 'K5' STO R7 R6 F2 'K6' STO

R2 R3 PZ2 'P1' STO R5 R7 PZ2 'P2' STO

R6 R7 PZ2 'P3' STO >>

'MAT'

<< E1 E1 K1 \* 0 E3 E2 K3 \* E2 K5 \* .6 .6 8 1

->LIST ->ARRY 8 IDN 'A' STO B2 1 + R1 \*

'A(1,8)' STO K2 NEG 'A(2,3)' STO P1 NEG

'A(2,8)' STO -1 'A(3,2)' STO B1 R3 \*

'A(3,7)' STO R4 'A(4,7)' STO K4 NEG

'A(5,6)' STO B1 1 + P2 \* NEG 'A(5,7)' STO

K6 NEG 'A(6,5)' STO B2 P3 \* NEG 'A(6,8)' STO

1 'A(7,4)' STO -1 'A(7,5)' STO 0 'A(7,7)' STO

1 'A(8,1)' STO -1 'A(8,2)' STO 0 'A(8,8)' STO

A / ARRY-> DROP >>

Step 4. Since this is a DC analysis, press DCAP when  
in the CKTAS directory:

In the stack see:	level 8:	3.860 (V1)
	level 7:	3.260 (V2)
	level 6:	3.258 (V3)
	level 5:	3.771 (V4)
	level 4:	3.171 (V5)
	level 3:	1.636 (V6)
	level 2:	1.229E-4 (Ib1)
	level 1:	0.001 (Ib2)

For an AC analysis of this circuit, it is suggested that the CE hybrid pi high frequency model be substituted for the simple linear DC model used here. A more accurate non-linear model can be created by using the diode equations in reference 5 and the Ebers-Moll models in reference 6.



Example 10 Non-linear Circuit.  
(Reference 3.)

Step 1. Purge KRV except for UP.

Step 2. Store the following component values:

R1 = 3 K  
R2 = 2 K  
R3 = 2 K  
R4 = 3 K  
R5 = 10 K  
R6 = 10 Meg (Initial value)  
R7 = 10 Meg " "  
E1 = +5 V  
E2 = -5 V

Step 3.

'SKF'

<< R1 R2 R6 F3 'K1' STO R2 R1 R6 F3 'K2' STO

R6 R1 R2 F3 'K3' STO R4 R3 R7 F3 'K4' STO

R3 R4 R7 F3 'K5' STO R7 R3 R4 'K6' STO

R6 R7 R5 F3 'K7' STO R7 R6 R5 F3 'K8' STO >>

'MAT'

<< 1 0 K3 NEG 0 1 K6 NEG K7 NEG K8 NEG 1

3 3 2 ->LIST ->ARRY E1 K1 \* E3 K2 \* +

E2 K4 \* E3 K5 \* + 0 3 1 ->LIST ->ARRY

SWAP / ARRY-> DROP >>

[Leave voltage vector  
on stack]

Step 4. In order to see the affect of a varying input voltage and to change the "diode" resistor values if forward or reversed biased, a different main program is required which will be labled VSWP for "voltage sweep". This main program is similar in structure to ACAP and is given below with comments

```

'VSWP'

-17 'EL' STO 15 'ER' STO          [Left & right sweep
ER EL - 2 / 1 + 1 ->LIST          limits]

0 0 R->C CON 'VOUT' STO

'VOUT' 1 1 ->LIST

EL ER FOR v v 'E3' STO

DCAP KRV

'V3' STO 'V2' STO 'V1' STO        [Store voltage
E3 V3 R->C                          vector]

IF V1 V3 - 0 < THEN 10

ELSE 10000000 END                  [Set diode resistance]

'R6' STO

IF V2 V3 - 0 > THEN 10

ELSE 10000000 END                  [Set diode resistance]

'R7' STO

CKTAS PUTI 2 STEP

EL 'BF' STO ER 'ND' STO           [For BPLT]

CLEAR VOUT ARRY-> DROP

KRV 10000000 DUP 'R6' STO         [Re-initialize diodes]

'R7' STO CKTAS >>

```

Step 4.

Level 1 = (15.000,6.245)

Level 8 = (1.000,0.001)

[Dead-space]

Level 16 = (-15.000,-6.245)

For plot, key in:

CLEAR, 7, space, -7, BPLT.


The dead-space portion of the limiter should be apparent.

## APPENDIX

### I. Ladder Network Analysis

All circuits analyzed so far have used matrices for the solution form. Ladder networks lend themselves to a more efficient solution form which will run faster.

Given the 4 L-section ladder network shown below:



Let  $B_0 = 1/Z_0$

$$B_4 = 1/Z_4 + 1/(Z_7 + 1/B_0)$$

$$B_4 = 1/Z_4 + 1/(Z_5 + 1/B_4)$$

$$B_2 = 1/Z_2 + 1/(Z_3 + 1/B_4)$$

Then

$$V_1/V_2 = 1 + Z_1 B_2$$

$$V_2/V_3 = 1 + Z_3 B_4$$

$$V_3/V_4 = 1 + Z_5 B_4$$

$$V_4/V_5 = 1 + Z_7 B_0$$

Finally  $V_1/V_5 =$

$$(1 + Z_1 B_2)(1 + Z_3 B_4)(1 + Z_5 B_4)(1 + Z_7 B_0)$$

Taking the inverse will give the transfer function  $V_5/V_1$ .

For output impedance  $Z_o$ :

$$Z_o = Z_6 // (Z_7 + Z_4 // (Z_5 + Z_4 // (Z_3 + Z_1 // Z_2)))$$

Or by chained fractions:

$$\text{Let } A_2 = 1/Z_2 + 1/Z_1$$


$$A_4 = 1/Z_4 + 1/(Z_3 + 1/A_2)$$

$$A_4 = 1/Z_4 + 1/(Z_5 + 1/A_4)$$

$$A_6 = 1/Z_6 + 1/(Z_7 + 1/A_4)$$

$$Z_o = 1/A_6$$

The ladder network shown below will be analyzed using the above expressions for the output voltage. The circuit is a model of a high frequency transformer. (Ref. 4)  
The topology is a 3 L-section ladder network.



Although SKF and MAT may now be inappropriately named, they are retained for the sake of consistency. The high frequency transformer model can be analyzed faster with the following routines than the heretofore standard matrix format:

Step 1. Purge KRV except UP.

Step 2. Store the following values in the KRV subdirectory:

R1 = 10	L6 = 2 mH
C2 = 20 pF	R7 = 1.5
R3 = 1.5	L8 = 1 uH
L4 = 1 uH	R9 = 1 K
R5 = 20 K	C10 = 20 pF

Step 3.

'SKF'

```
<< C2 C 'Z2' STO R3 L4 SRL 'Z3' STO R5 L6 PRL 'Z4' STO  
R7 L8 SRL 'Z5' STO R9 C10 PRC 'Z6' STO >>
```

'MAT'

```
<< Z6 Z5 + INV Z4 INV + 'B4' STO B4 INV Z3 + INV  
Z2 INV + 'B2' STO Z5 Z6 / 1 + B4 Z3 * 1 + *  
B2 R1 * 1 + * INV >>
```

Step 4.

A "6 5 2" frequency sweep will show both the low and high frequency response:

6, space, 5, space, 2, press ACAP.

Level 31:	(2.000,-19.296)	(Low frequency response)
Level 16:	(5.000,-0.122)	(Midband response)
Level 4:	(7.400,8.849)	(Resonance peak)
Level 1:	(8.000,-23.496)	(High frequency rolloff)


Plot a max of 10 dB, and a min of -25 dB:

CLEAR, 10, space, -25, press BPLT.

The resonance peak is followed by a sharp rolloff.

## II. Building Branch Impedances with the HP-28S

Branch impedances other than the simple series and parallel RC or RL given by the subprograms on page 22 are easy to construct. For example, for the branch impedance Z1 shown below




the HP-28S sequence is

```
R1 C2 PRC R2 C1 SRC R3 L1 SRL PZ3 R4 C3 SRC +
'Z1' STO.
```

## III. Floating Voltage Sources

Floating voltage sources are sometimes required for diode and transistor models where the value of the voltage source in series with a resistor is about 0.6V. Whatever the purpose, they are analyzed as shown in the example below:



Step 1.

$$V_1 = E_1 K_1 + V_2 K_2 + (V_2 - E_3) K_3 \quad \text{or}$$

$$V_1 = E_1 K_1 + V_2 (K_2 + K_3) - E_3 K_3$$

$$V_2 = E_2 K_4 + V_1 K_5 + (V_1 + E_3) K_6 \quad \text{or}$$

$$V_2 = E_2 K_4 + V_1 (K_5 + K_6) - E_3 K_6$$

Step 3.

$$\begin{vmatrix} 1 & -(K_2 + K_3) \\ -(K_5 + K_6) & 1 \end{vmatrix} \begin{vmatrix} V_1 \\ V_2 \end{vmatrix} = \begin{vmatrix} E_1 K_1 - E_3 K_3 \\ E_2 K_4 + E_3 K_6 \end{vmatrix}$$

where the intermediate step 2. is skipped. Note that node  $V_1$  "sees" a voltage of  $V_2 - E_3$  when "looking at" node  $V_2$  via impedance  $Z_3$ . Conversely, node  $V_2$  "sees" a voltage of  $V_1 + E_3$  when "looking at" node  $V_1$  via  $Z_6$ . Hence the polarity of the floating source must be observed when writing the node equations.



#### IV. Designing with K Factors


The following is an example of how K factors can be used in circuit design:

Given the  $N = 4$  star network shown below, determine the required resistor values such that

$$V_o = V_1 K_1 + V_2 K_2 + V_3 K_3 + V_4 K_4$$

$$= V_1(0.1) + V_2(0.2) + V_3(0.3) + V_4(0.4)$$

for any values of  $V_1$  thru  $V_4$ . (Note that  $K_1 + K_2 + K_3 + K_4$  must = 1.)



One solution method would be to generate a set of four simultaneous equations from Kirchoff's Current Law or Kirchoff's Voltage Law, for the four unknown resistor values. However, using K factors allows the simultaneous equations to be avoided:

Let  $R_1 = 1K$ , then

$$R_2 = R_1 K_1 / K_2 = 1000(0.1) / 0.2 = 500 \text{ ohms.}$$

$$R_3 = R_1 K_1 / K_3 = 100 / 0.3 = 333 \text{ ohms.}$$

$$R_4 = R_1 K_1 / K_4 = 100 / 0.4 = 250 \text{ ohms.}$$

For an  $N = 3$  branch, assume the design requirements are:

$$K_1 = 0.2, K_2 = 0.25, \text{ \& } K_3 = 0.15.$$

In this case,  $K_1 + K_2 + K_3 = 0.6 < 1$ , and we must provide a fourth branch with  $V_4 = 0$  and

$$K_4 = 1 - 0.6 = 0.4.$$

Again letting  $R_1$  be 1K:

$$R_2 = 1000(0.2)/.25 = 800 \text{ ohms},$$

$$R_3 = 200/0.15 = 1333 \text{ ohms},$$

$$R_4 = 200/0.4 = 500 \text{ ohms}.$$

The star or summing network is useful where the output  $V_o$  is connected to a high impedance such as non-inverting op amp or comparator inputs.

References:

1. Pulse, Digital, and Switching Waveforms, Millman & Taub, 1965, p. 7.
2. Operational Amplifiers, Design and Applications, Graeme, Tobey, & Huelsman, 1971, p. 202.
3. Introduction to Operational Amplifier Theory and Applications, Wait, Huelsman, & Korn, 1975, p. 147.
4. Reference Data for Radio Engineers, 5th Ed., ITT/Sams, p. 12-1.
5. Algorithms for RPN Calculators, Ball, 1978, p. 272.
6. Modeling The Bipolar Transistor, Getreu, 1976, Tektronix.

